



Subtracting this from p itself will remove these three bits, leaving an integer value between 0 and 31, representing the bottom (rightmost) five bits of p . The value of $FN\ a(p)$ will therefore always be defined with the top three bits set to zero, whatever the value of the top three bits of p .

For example, if the letter V is pressed, port 65278 will contain:

X	X	X	0	1	1	1	1
---	---	---	---	---	---	---	---

$FN\ a(IN(65278))$ will therefore return the value:

0 0 0 0 1 1 1 1 = 15

If both Caps Shift and the letter V are pressed simultaneously, the port will contain:

X	X	X	0	1	1	1	0
---	---	---	---	---	---	---	---

and $FN\ a(IN(65278))$ will therefore return the value 14. If no keys are pressed at all, the port will return the value 31 (00011111).

Try this program:

```

10 REM Print out the masked value of the ports
20 REM Define the masking function
30 DEF FN a(p)=p - INT(p/32)*32
40 REM Print out the ports
50 PRINT AT 1,1;"Port           Masked value"
60 PRINT "32766",FN a(IN(32766))
70 PRINT "49150",FN a(IN(49150))
80 PRINT "57342",FN a(IN(57342))
90 PRINT "61438",FN a(IN(61438))
100 PRINT "63486",FN a(IN(63486))
110 PRINT "64510",FN a(IN(64510))
120 PRINT "65022",FN a(IN(65022))
130 PRINT "65278",FN a(IN(65278))
140 FOR i=1 TO 250: NEXT i
150 CLS
160 GO TO 50

```

After entering RUN , try pressing some keys and notice how the masked values of the ports change. Try to predict the values that will be displayed for different keys and combinations of keys. If you press the cursor keys (5, 6, 7 and 8) you should get the following masked values on ports 61438 and 63486 respectively (which we will call port A and port B)

	Port A (port 61438)	Port B (port 63486)
←	31	15
↓	15	15
↑	23	15
→	27	15

If you press the Caps Shift key, you should get the masked value of 30 for port 65278 (which we will refer to as port C).

Consider the simple graphics program given here. This program will read the cursor keys to draw horizontal, vertical and diagonal lines and, if the Caps Shift key is pressed, move a 'graphics pen' without drawing a line. Diagonal lines are drawn by pressing two cursor keys at once.

In the program, line 30 defines a function ($FN(a)$) to mask off the top three bits, as before. The horizontal position of the graphics pen is represented by x , and y is its vertical co-ordinate. Line 50 sets the pen's initial position at the centre of the screen.

Lines 70 and 80 read in the ports mapped to the cursor keys. The half-row from 6 to 0 includes three of the cursor keys and maps to port 61438 (port A). The half-row 5 to 1 contains the right arrow cursor key and maps to port 63486 (port B). Line 90 reads the Caps Shift key — the half row from V to Caps Shift is mapped to port 65278 (port C).

Lines 110 to 180 test for the eight 'legal' combinations of cursor keys, and the x,y position of the graphics pen is adjusted accordingly:

Line 110 tests for ↑
 Line 120 tests for ↑ and →
 Line 130 tests for →
 Line 140 tests for ↓ and →
 Line 150 tests for ↓
 Line 160 tests for ↓ and ←
 Line 170 tests for ←
 Line 180 tests for ↑ and ←

Lines 200 to 240 ensure that the pen does not move off the edge of the screen. Finally, if the Caps Shift has not been pressed, line 250 will plot the graphics pen's new position on the screen. Line 260 then loops back to repeat the whole process again.

Shifty Business

The program employs the techniques discussed for detecting multiple keypresses: the unshifted arrow keys allow you to draw horizontal, vertical and diagonal lines (diagonal=horizontal+vertical), while the shifted keys produce the same cursor movements without screen plotting. The same techniques will be used to better effect in a future article

Detecting Multiple Keypresses

```

20 REM Set Up Function to Mask Top Three Bits
30 DEF FN a(p)=p-INT (p/32)*32
40 REM Initialise Pen Position
50 LET x=127: LET y=35
60 REM Read Ports and Mask Off Top Three Bits
70 LET portA=FN a(IN 61438)
80 LET portB=FN a(IN 63436)
90 LET portC=FN a(IN 65278)
100 REM Alter Pen Position Depending on Which Key
    Has Been Pressed
110 IF portA=23 AND portB=31 THEN LET y=y+1
120 IF portA=19 AND portB=31 THEN LET x=x+1:
    LET y=y+1
130 IF portA=27 AND portB=31 THEN LET x=x+1
140 IF portA=11 AND portB=31 THEN LET x=x+1:
    LET y=y-1
150 IF portA=15 AND portB=31 THEN LET y=y-1
160 IF portA=15 AND portB=15 THEN LET x=x-1:
    LET y=y-1
170 IF portA=31 AND portB=15 THEN LET x=x-1
180 IF portA=23 AND portB=15 THEN LET y=y+1:
    LET x=x-1
190 REM Stop Pen Going Off The Screen
200 IF x<0 THEN LET x=0
210 IF x>255 THEN LET x=255
220 IF y<0 THEN LET y=0
230 IF y>175 THEN LET y=175
240 REM Plot Point if CAPS SHIFT Not Pressed
250 IF portC=31 THEN PLOT x,y
260 GO TO 70

```